# $Id: PKGBUILD 175921 2013-01-25 01:07:36Z bisson $
# Contributor: dorphell <dorphell@archlinux.org>
# Maintainer: Gaetan Bisson <bisson@archlinux.org>

pkgname=libao
pkgver=1.1.0
pkgrel=3
pkgdesc='Cross-platform audio output library and plugins'
url='http://xiph.org/ao/'
arch=('i686' 'x86_64')
license=('GPL')
options=('!libtool')
depends=('alsa-lib')
makedepends=('libpulse')
backup=('etc/libao.conf')
source=("http://downloads.xiph.org/releases/ao/${pkgname}-${pkgver}.tar.gz"
        'libao.conf')
sha1sums=('9301bc4886f170c7122ab62677fb71cf001c04fd'
          '603f5e6715e7e50e1c8e8c1935c45a897c46e9af')

conflicts=('libao-pulse')
provides=("libao-pulse=${pkgver}-${pkgrel}")
replaces=('libao-pulse')

build() {
	cd "${srcdir}/${pkgname}-${pkgver}"

	./configure \
		--prefix=/usr \
		--enable-alsa-mmap \

	make
}

package() {
	cd "${srcdir}/${pkgname}-${pkgver}"
	make DESTDIR="${pkgdir}" install
	install -Dm644 "${srcdir}"/libao.conf "${pkgdir}"/etc/libao.conf
}
